Multiple scales analysis of the van der Pol oscillator

```
> VDP:=diff(x(t),t,t)-epsilon*diff(x(t),t)*(1-x(t)^2)+omega[0]^2*x(t);
```

$$VDP := \left(\frac{d^2}{dt^2}x(t)\right) - \varepsilon\left(\frac{d}{dt}x(t)\right)(1 - x(t)^2) + \omega_0^2\, x(t)$$

Make the change of variable $T = \Omega\, t$

```
> S1:=[T=Omega*t,x(t)=y(T)];
```

$$S1 := [T = \Omega\, t, x(t) = y(T)]$$

```
> VDP1:=convert(subs(t=solve(S1[1],t),subs(Omega=solve(S1[1],Omega),eval
  (subs(subs(S1[1],S1[2]),VDP)))),diff);
```

$$VDP1 := \left(\frac{d^2}{dT^2}y(T)\right)\Omega^2 - \varepsilon\left(\frac{d}{dT}y(T)\right)\Omega\,(1 - y(T)^2) + \omega_0^2\, y(T)$$

Expand Ω and $y(\tau)$ in powers of ε and substitute in the differential equation

```
> S2:=[Omega=omega[0]+epsilon*omega[1]+epsilon^2*omega[2],y(T)=y[0](T)+e
  psilon*y[1](T)+epsilon^2*y[2](T)];
```

$$S2 := [\Omega = \omega_0 + \varepsilon\,\omega_1 + \varepsilon^2\,\omega_2, y(T) = y_0(T) + \varepsilon\, y_1(T) + \varepsilon^2\, y_2(T)]$$

```
> VDP2:=convert(series(subs(simplify(subs(S2,VDP1))),epsilon,3),diff);
```

$$VDP2 := \left(\omega_0^2\, y_0(T) + \left(\frac{d^2}{dT^2}y_0(T)\right)\omega_0^2\right) + \left(-\left(\frac{d}{dT}y_0(T)\right)\omega_0 + \omega_0^2\, y_1(T) + 2\left(\frac{d^2}{dT^2}y_0(T)\right)\omega_0\,\omega_1 + \left(\frac{d}{dT}y_0(T)\right)\omega_0\, y_0(T)^2 + \left(\frac{d^2}{dT^2}y_1(T)\right)\omega_0^2\right)\varepsilon + \left(2\left(\frac{d}{dT}y_0(T)\right)\omega_0\, y_0(T)\, y_1(T) - \left(\frac{d}{dT}y_0(T)\right)\omega_1 + \left(\frac{d^2}{dT^2}y_0(T)\right)\omega_1^2 - \left(\frac{d}{dT}y_1(T)\right)\omega_0 + \left(\frac{d^2}{dT^2}y_2(T)\right)\omega_0^2 + 2\left(\frac{d^2}{dT^2}y_1(T)\right)\omega_0\,\omega_1 + \left(\frac{d}{dT}y_1(T)\right)\omega_0\, y_0(T)^2 + 2\left(\frac{d^2}{dT^2}y_0(T)\right)\omega_0\,\omega_2 + \omega_0^2\, y_2(T) + \left(\frac{d}{dT}y_0(T)\right)\omega_1\, y_0(T)^2\right)\varepsilon^2 + O(\varepsilon^3)$$

Note that the above equation is only valid up to order ε^2 since the expansions for ω and $y(T)$ were truncated at order 3.

Solve the differential equations obtained at each order

Order ε^0

```
> VDP0:=convert(subs(epsilon=0,VDP2),diff);
```

$$VDP0 := \omega_0^2\, y_0(T) + \left(\frac{d^2}{dT^2} y_0(T)\right) \omega_0^2$$

```
> E1:=y[0](T)=A*cos(T+phi);
```

$$E1 := y_0(T) = A\cos(T+\varphi)$$

```
> simplify(eval(subs(E1,VDP0)));
```

$$0$$

Order ε

```
> coeff(VDP2,epsilon);
```

$$-\left(\frac{d}{dT} y_0(T)\right)\omega_0 + \omega_0^2\, y_1(T) + 2\left(\frac{d^2}{dT^2} y_0(T)\right)\omega_0\,\omega_1 + \left(\frac{d}{dT} y_0(T)\right)\omega_0\, y_0(T)^2 + \left(\frac{d^2}{dT^2} y_1(T)\right)\omega_0^2$$

Substitute solution at order ε^0

```
> E2:=convert(collect(simplify(eval(subs(E1,coeff(VDP2,epsilon)))),
  [diff(y[1](T),T,T),y[1](T)]),diff);
```

$$E2 := \left(\frac{d^2}{dT^2} y_1(T)\right)\omega_0^2 + \omega_0^2\, y_1(T) - \omega_0\left(-A\sin(T+\varphi) + 2A\cos(T+\varphi)\,\omega_1 + A^3\sin(T+\varphi)\cos(T+\varphi)^2\right)$$

Solve for $y_1(t)$

```
> E3:=collect(simplify(subs(_C1=0,_C2=0,dsolve(E2,y[1](T)))),T,comb
  ine);
```

$$E3 := y_1(T) = \frac{\left(\frac{1}{2}A\cos(T+\varphi) + A\sin(T+\varphi)\,\omega_1 - \frac{1}{8}A^3\cos(T+\varphi)\right)T}{\omega_0} + \frac{-8A\sin(T+\varphi) + 16A\cos(T+\varphi)\,\omega_1 - A^3\sin(3T+3\varphi) + 2A^3\sin(T+\varphi)}{32\,\omega_0}$$

Note that $y_1(T)$ diverges, unless the term proportional to T is zero. This gives us a **solvability condition** for the differntial equation for $y_1(T)$

Solvability condition

```
> E4:=collect(A*omega[1]*sin(T+phi)+1/2*A*cos(T+phi)-1/8*A^3*cos(T+
  phi),[cos(T+phi),sin(T+phi)])=0;
```

$$E4 := \left(\frac{1}{2}A - \frac{1}{8}A^3\right)\cos(T+\varphi) + A\sin(T+\varphi)\,\omega_1 = 0$$

Find the values of A and ω_1 which satisfy the solvability condition

```
> E5:={coeff(op(1,E4),cos(T+phi))=0,coeff(op(1,E4),sin(T+phi))=0};
```

$$E5 := \left\{ A\,\omega_1 = 0, \frac{1}{2} A - \frac{1}{8} A^3 = 0 \right\}$$

```
> solve(E5,{A,omega[1]});
```

$$\{A = 0, \omega_1 = \omega_1\}, \{A = 2, \omega_1 = 0\}, \{A = -2, \omega_1 = 0\}$$

From now on, we set $A = 2$ **and** $\omega_1 = 0$.

Re-write the solutions obtained at orders ε^0 ***and*** ε

```
> E6:=[subs(A=2,E1),subs(A=2,collect(subs(omega[1]=0,E3),[cos(T+phi),sin(T+phi)])),omega[1]=0];
```

$$E6 := \left[y_0(T) = 2\cos(T+\varphi), y_1(T) = -\frac{\sin(3\,T + 3\,\varphi)}{4\,\omega_0}, \omega_1 = 0 \right]$$

Order ε^2

Equation for $y_2(T)$

```
> E7:=simplify(eval(subs(E6,coeff(VDP2,epsilon^2))));
```

$$E7 := 2\sin(T+\varphi)\cos(T+\varphi)\sin(3\,T+3\,\varphi) + \frac{3}{4}\cos(3\,T+3\,\varphi) + \left(\frac{d^2}{dT^2} y_2(T)\right)\omega_0^2 - 3\cos(3\,T+3\,\varphi)\cos(T+\varphi)^2 - 4\cos(T+\varphi)\,\omega_0\,\omega_2 + \omega_0^2\,y_2(T)$$

Solution. Note the secular terms in T

```
> E8:=collect(subs(_C1=0,_C2=0,dsolve(E7,y[2](T))),T,combine);
```

$$E8 := y_2(T) = \frac{(16\sin(T+\varphi)\,\omega_0\,\omega_2 + \sin(T+\varphi))\,T}{8\,\omega_0^2} + \frac{96\cos(T+\varphi)\,\omega_0\,\omega_2 + 6\cos(T+\varphi) - 5\cos(5\,T+5\,\varphi) - 9\cos(3\,T+3\,\varphi)}{96\,\omega_0^2}$$

Solvability conditions - set the term in T ***to*** 0 ***and solve for*** ω_2

```
> E9:=1/8*(16*sin(T+phi)*omega[0]*omega[2]+sin(T+phi))/omega[0]^2=0;
```

$$E9 := \frac{16\sin(T+\varphi)\,\omega_0\,\omega_2 + \sin(T+\varphi)}{8\,\omega_0^2} = 0$$

```
> E10:omega[2]=solve(E9,omega[2]);
```

$$\omega_2 = -\frac{1}{16\,\omega_0}$$